

NAG Fortran Library Routine Document

F08HQF (CHBEVD/ZHBEVD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

Warning. The specification of the parameters LRWORK and LIWORK changed at Mark 20 in the case where JOB = 'V' and $N > 1$: the minimum dimension of the array RWORK has been reduced whereas the minimum dimension of the array IWORK has been increased.

1 Purpose

F08HQF (CHBEVD/ZHBEVD) computes all the eigenvalues, and optionally all the eigenvectors, of a complex Hermitian band matrix. If the eigenvectors are requested, then it uses a divide and conquer algorithm to compute eigenvalues and eigenvectors. However, if only eigenvalues are required, then it uses the Pal–Walker–Kahan variant of the QL or QR algorithm.

2 Specification

```

SUBROUTINE F08HQF(JOB, UPLO, N, KD, AB, LDAB, W, Z, LDZ, WORK, LRWORK,
1                RWORK, LRWORK, IWORK, LIWORK, INFO)
ENTRY          chbevd (JOB, UPLO, N, KD, AB, LDAB, W, Z, LDZ, WORK, LRWORK,
1                RWORK, LRWORK, IWORK, LIWORK, INFO)
INTEGER        N, KD, LDAB, LDZ, LRWORK, LRWORK, IWORK(*), LIWORK,
1                INFO
real          W(*), RWORK(*)
complex      AB(LDAB,*), Z(LDZ,*), WORK(*)
CHARACTER*1    JOB, UPLO

```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine computes all the eigenvalues, and optionally all the eigenvectors, of a complex Hermitian band matrix A . In other words, it can compute the spectral factorization of A as

$$A = ZAZ^H,$$

where A is a real diagonal matrix whose diagonal elements are the eigenvalues λ_i , and Z is the (complex) unitary matrix whose columns are the eigenvectors z_i . Thus

$$Az_i = \lambda_i z_i, \quad i = 1, 2, \dots, n.$$

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: JOB – CHARACTER*1

Input

On entry: indicates whether eigenvectors are computed as follows:

if JOB = 'N', only eigenvalues are computed;

if JOB = 'V', eigenvalues and eigenvectors are computed.

Constraint: JOB = 'N' or 'V'.

- 2: UPLO – CHARACTER*1 *Input*
On entry: indicates whether the upper or lower triangular part of A is stored as follows:
 if UPLO = 'U', the upper triangular part of A is stored;
 if UPLO = 'L', the lower triangular part of A is stored.
Constraint: UPLO = 'U' or 'L'.
- 3: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 4: KD – INTEGER *Input*
On entry: k , the number of super-diagonals of the matrix A if UPLO = 'U', or the number of sub-diagonals if UPLO = 'L'.
Constraint: $KD \geq 0$.
- 5: AB(LDAB,*) – **complex** array *Input/Output*
Note: the second dimension of the array AB must be at least $\max(1, N)$.
On entry: the upper or the lower triangle of the n by n Hermitian band matrix A , stored in the first $KD + 1$ rows of the array AB. More precisely, the j th column of A is stored in the j th column of the array AB as follows:
 if UPLO = 'U', $AB(KD + 1 + i - j, j) = a_{ij}$ for $\max(1, j - KD) \leq i \leq j$;
 if UPLO = 'L', $AB(1 + i - j, j) = a_{ij}$ for $j \leq i \leq \min(n, j + KD)$.
On exit: A is overwritten by the values generated during the reduction to tridiagonal form. If UPLO = 'U', the first superdiagonal and the diagonal of the tridiagonal matrix are returned in rows KD and $KD + 1$ of the array AB, respectively, and if UPLO = 'L' then the diagonal and the first subdiagonal of the tridiagonal matrix are returned in the first two rows of the array AB.
- 6: LDAB – INTEGER *Input*
On entry: the first dimension of the array AB as declared in the (sub)program from which F08HQF (CHBEVD/ZHBEVD) is called.
Constraint: $LDAB \geq KD + 1$.
- 7: W(*) – **real** array *Output*
Note: the dimension of the array W must be at least $\max(1, N)$.
On exit: the eigenvalues of the matrix A in ascending order.
- 8: Z(LDZ,*) – **complex** array *Output*
Note: the second dimension of the array Z must be at least $\max(1, N)$ if JOB = 'V', and at least 1 if JOB = 'N'.
On exit: if JOB = 'V', Z is overwritten by the unitary matrix Z which contains the eigenvectors of A . The i th column of Z contains the eigenvector which corresponds to the eigenvalue $W(i)$.
 If JOB = 'N', Z is not referenced.

- 9: LDZ – INTEGER *Input*
On entry: the first dimension of the array Z as declared in the (sub)program from which F08HQF (CHBEVD/ZHBEVD) is called.
Constraints:

$$\text{LDZ} \geq \max(1, N) \text{ if JOB} = \text{'V'},$$

$$\text{LDZ} \geq 1 \text{ if JOB} = \text{'N'}.$$
- 10: WORK(*) – **complex** array *Workspace*
Note: the dimension of the array WORK must be at least $\max(1, \text{LWORK})$.
On exit: if INFO = 0, the real part of WORK(1) contains the required minimal size of LWORK.
- 11: LWORK – INTEGER *Input*
On entry: the dimension of the array WORK as declared in the (sub)program from which F08HQF (CHBEVD/ZHBEVD) is called, unless LWORK = -1, in which case a workspace query is assumed and the routine only calculates the minimum dimension of WORK.
Constraints:

$$\text{if } N \leq 1, \text{LWORK} \geq 1 \text{ or } \text{LWORK} = -1,$$

$$\text{if JOB} = \text{'N'} \text{ and } N > 1, \text{LWORK} \geq N \text{ or } \text{LWORK} = -1,$$

$$\text{if JOB} = \text{'V'} \text{ and } N > 1, \text{LWORK} \geq 2 \times N^2 \text{ or } \text{LWORK} = -1.$$
- 12: RWORK(*) – **real** array *Workspace*
Note: the dimension of the array RWORK must be at least $\max(1, \text{LRWORK})$.
On exit: if INFO = 0, RWORK(1) contains the required minimal size of LRWORK.
- 13: LRWORK – INTEGER *Input*
On entry: the dimension of the array RWORK as declared in the (sub)program from which F08HQF (CHBEVD/ZHBEVD) is called, unless LRWORK = -1, in which case a workspace query is assumed and the routine only calculates the minimum dimension of RWORK.
Constraints:

$$\text{if } N \leq 1, \text{LRWORK} \geq 1 \text{ or } \text{LRWORK} = -1,$$

$$\text{if JOB} = \text{'N'} \text{ and } N > 1, \text{LRWORK} \geq N \text{ or } \text{LRWORK} = -1,$$

$$\text{if JOB} = \text{'V'} \text{ and } N > 1, \text{LRWORK} \geq 2 \times N^2 + 5 \times N + 1 \text{ or } \text{LRWORK} = -1.$$
- 14: IWORK(*) – INTEGER array *Workspace*
Note: the dimension of the array IWORK must be at least $\max(1, \text{LIWORK})$.
On exit: if INFO = 0, IWORK(1) contains the required minimal size of LIWORK.
- 15: LIWORK – INTEGER *Input*
On entry: the dimension of the array IWORK as declared in the (sub)program from which F08HQF (CHBEVD/ZHBEVD) is called, unless LIWORK = -1, in which case a workspace query is assumed and the routine only calculates the minimum dimension of IWORK.
Constraints:

$$\text{if JOB} = \text{'N'} \text{ or } N \leq 1, \text{LIWORK} \geq 1 \text{ or } \text{LIWORK} = -1,$$

$$\text{if JOB} = \text{'V'} \text{ and } N > 1, \text{LIWORK} \geq 5 \times N + 3 \text{ or } \text{LIWORK} = -1.$$
- 16: INFO – INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO = i , the algorithm failed to converge; i indicates the number of elements of an intermediate tridiagonal form which did not converge to zero.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $A + E$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the *machine precision*.

8 Further Comments

The real analogue of this routine is F08HCF (SSBEVD/DSBEVD).

9 Example

To compute all the eigenvalues and eigenvectors of the Hermitian band matrix A , where

$$A = \begin{pmatrix} 1.0 + 0.0i & 2.0 + 1.0i & 3.0 + 1.0i & 0.0 + 0.0i & 0.0 + 0.0i \\ 2.0 - 1.0i & 2.0 + 0.0i & 3.0 + 2.0i & 4.0 + 2.0i & 0.0 + 0.0i \\ 3.0 - 1.0i & 3.0 - 2.0i & 3.0 + 0.0i & 4.0 + 3.0i & 5.0 + 3.0i \\ 0.0 + 0.0i & 4.0 - 2.0i & 4.0 - 3.0i & 4.0 + 0.0i & 5.0 + 4.0i \\ 0.0 + 0.0i & 0.0 + 0.0i & 5.0 - 3.0i & 5.0 + 0.0i & 5.0 + 0.0i \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F08HQF Example Program Text.
*      Mark 20 Revised. NAG Copyright 2001.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5, NOUT=6)
      INTEGER          NMAX, LDZ, LDAB
      PARAMETER        (NMAX=8, LDZ=NMAX, LDAB=NMAX)
      INTEGER          LWORK, LIWORK, LRWORK
      PARAMETER        (LWORK=2*NMAX*NMAX, LIWORK=5*NMAX+3,
+                     LRWORK=2*NMAX*NMAX+5*NMAX+1)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, KD, N
      CHARACTER        JOB, UPLO
*      .. Local Arrays ..
      complex         AB(LDAB, NMAX), WORK(LWORK), Z(LDZ, NMAX)
      real            RWORK(LRWORK), W(NMAX)
      INTEGER          IWORK(LIWORK)
*      .. External Subroutines ..
      EXTERNAL         X04DAF, chbevd
*      .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F08HQF Example Program Results'
```

```

*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KD
      IF (N.LE.NMAX) THEN
*
*          Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
          DO 20 I = 1, N
              READ (NIN,*) (AB(KD+1+I-J,J),J=I,MIN(N,I+KD))
20          CONTINUE
      ELSE IF (UPLO.EQ.'L') THEN
          DO 40 I = 1, N
              READ (NIN,*) (AB(1+I-J,J),J=MAX(1,I-KD),I)
40          CONTINUE
      END IF
*
      READ (NIN,*) JOB
*
*      Calculate all the eigenvalues and eigenvectors of A
*
      CALL chbevd(JOB,UPLO,N,KD,AB,LDAB,W,Z,LDZ,WORK,LWORK,RWORK,
+                LRWORK,IWORK,LIWORK,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.GT.0) THEN
          WRITE (NOUT,*) 'Failure to converge.'
      ELSE
*
*          Print eigenvalues and eigenvectors
*
          WRITE (NOUT,*) 'Eigenvalues'
          DO 60 I = 1, N
              WRITE (NOUT,99999) I, W(I)
60          CONTINUE
          WRITE (NOUT,*)
          IFAIL = 0
*
          CALL X04DAF('General',' ',N,N,Z,LDZ,'Eigenvectors',IFAIL)
*
          END IF
      END IF
      STOP
*
99999 FORMAT (3X,I5,5X,2F8.4)
      END

```

9.2 Program Data

F08HQF Example Program Data

```

5 2                                     :Values of N and KD
'L'                                     :Value of UPLO
(1.0, 0.0)
(2.0, 1.0) (2.0, 0.0)
(3.0, 1.0) (3.0, 2.0) (3.0, 0.0)
              (4.0, 2.0) (4.0, 3.0) (4.0, 0.0)
              (5.0, 3.0) (5.0, 4.0) (5.0, 0.0) :End of matrix A
'V'                                     :Value of JOB

```

9.3 Program Results

F08HQF Example Program Results

Eigenvalues

1	-6.4185
2	-1.4094
3	1.4421
4	4.4856
5	16.9002

Eigenvectors

	1	2	3	4	5
1	-0.2591 0.0000	0.6367 0.0000	0.4516 0.0000	0.5503 0.0000	0.1439 0.0000
2	0.0245 0.4344	-0.2578 0.2413	-0.3029 -0.4402	0.4785 0.2759	0.3060 0.0411
3	0.5159 -0.1095	-0.3039 -0.3481	0.3160 0.2978	0.2128 0.0465	0.4681 0.2306
4	0.0004 -0.5093	0.3450 -0.0832	-0.4088 -0.3213	-0.1707 0.0200	0.4098 0.3832
5	-0.4333 0.1353	-0.2469 0.2634	0.0204 0.2262	0.0175 -0.5611	0.1819 0.5136
